# How to FullSimplify/Simplify an inequality while keep a variable isolated

I have an inequality as follows

2^(1/2 (1 + 1/n)) > 0 &&
t <= (2^(1/2 (-1 - 1/n) + n/2) n^(1 + 1/(2 n)) \[Pi]^((1/2)/n))/E


I want to simplify the 2^(1/2 (1 + 1/n)) > 0 to True using assumption that n > 0.

However, if I do the following,

2^(1/2 (1 + 1/n)) > 0 &&
t <= (2^(1/2 (-1 - 1/n) + n/2) n^(1 + 1/(2 n)) \[Pi]^((1/2)/n))/E //
FullSimplify[#, n > 0] &


I end up with

2^(1/2 (1 + 1/n)) E t <= 2^(n/2) n^(1 + 1/(2 n)) \[Pi]^((1/2)/n)


But I want to keep the t on one side of inequality. How can I do that.

Note the example is a bit simplified. I have a much complicated expression which I get from Reduce which I want to simplify, while keep t isolated on one side of inequalities.

• Let us observe that 2^(1/2 (1 + 1/n))  is positive at any n. Therefore, you can safely replace your inequality by t <= (2^(1/2 (-1 - 1/n) + n/2) n^(1 + 1/(2 n)) \[Pi]^((1/2)/n))/E &&n > 0, and work with this one. You did not describe, though, what are you expecting to get from it? Limitations on what variable do you want to obtain? Oct 14 at 11:30

Clear["Global*"]

expr = 2^(1/2 (1 + 1/n)) > 0 &&
t <= (2^(1/2 (-1 - 1/n) + n/2) n^(1 + 1/(2 n)) π^((1/2)/n))/E;

expr2 = ReplacePart[expr, 1 -> Simplify[expr[[1]], n > 0]]

(* t <= (2^(1/2 (-1 - 1/n) + n/2) n^(1 + 1/(2 n)) π^((1/2)/n))/E *)

• Why do we need to clear "*"? Oct 14 at 4:59
• It clears the Global namespace to ensure that there are no old definitions laying around (i.e., for n or t) that might interfere with the subsequent code. Oct 14 at 5:03

Or

expr = 2^(1/2 (1 + 1/n)) > 0 &&
t <= (2^(1/2 (-1 - 1/n) + n/2) n^(1 + 1/(2 n)) \[Pi]^((1/2)/n))/E;

expr // Refine[#, Assumptions -> n > 0] &

(*   t <= (2^(1/2 (-1 - 1/n) + n/2) n^(1 + 1/(2 n)) \[Pi]^(1/(2 n)))/E   *)


And with definite n

expr // Refine[#, n == 2] &

(*   t <= (2 Sqrt[2] \[Pi]^(1/4))/E   *)